FONCTIONS DE *n* VARIABLES RÉELLES : SOLUTIONS DES EXERCICES

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Exercice M1

Enoncé

Soit la fonction $f(x, y) = \frac{x^\alpha y}{x^2+y^2}$.

1. Calculer les dérivées partielles de *f*.

2. Calculer les élasticités de *f* par rapport à *x* et à *y*.

Solution

La fonction *f* est définie sur $\mathbb{R}^2 - \{(0, 0)\}$.

```
> restart;
  f:=(x,y)->(x^alpha*y)/(x^2+y^2);
```

$$f := (x, y) \to \frac{x^\alpha y}{x^2+y^2} \quad \textbf{(1.1)}$$

1. Le calcul des dérivées partielles ne présente aucune difficulté, que ce soit pour récupérer une expression ou une fonction.

```
> Xpfx:=diff(f(x,y),x);#expression dérivée partielle par rapport à x
  Xpfy:=diff(f(x,y),y);#expression dérivée partielle par rapport à y
  fpx:=D[1](f);#fonction dérivée partielle par rapport à x
  fpy:=D[2](f);#fonction dérivée partielle par rapport à y
```

$$Xpfx := \frac{x^\alpha \alpha y}{x(x^2+y^2)} - \frac{2 x^\alpha y x}{(x^2+y^2)^2}$$

$$Xpfy := \frac{x^\alpha}{x^2+y^2} - \frac{2 x^\alpha y^2}{(x^2+y^2)^2}$$

$$fpx := (x, y) \to \frac{x^\alpha \alpha y}{x(x^2+y^2)} - \frac{2 x^\alpha y x}{(x^2+y^2)^2}$$

$$fpy := (x, y) \to \frac{x^\alpha}{x^2+y^2} - \frac{2 x^\alpha y^2}{(x^2+y^2)^2} \quad \textbf{(1.2)}$$

2. L'élasticité de *f* par rapport à la variable *x* est, par définition :

$$e(f;x) = \frac{x \frac{\partial f}{\partial x}(x,y)}{f(x,y)}.$$

```
> Efx:=simplify(x*Xpfx/f(x,y));#élasticité par rapport à x
  Efy:=simplify(y*Xpfy/f(x,y));#élasticité par rapport à y
```

$$Efx := \frac{\alpha x^2 + \alpha y^2 - 2 x^2}{x^2 + y^2}$$

$$Efy := \frac{x^2 - y^2}{x^2 + y^2} \qquad (1.3)$$

Exercice M2

Enoncé

Calculer une valeur approchée de $x^{\alpha} y^{1-\alpha}$ au point $(x, y) = (10.1, 9.95)$.

Solution

La question est floue au sens où on ne connaît pas vraiment la précision demandée. Généralement on se contente d'une approximation affine, qui correspond en Maple à l'utilisation directe de `mtaylor` à l'ordre 1.

```
> restart;
  f:=(x,y)->x^alpha*y^(1-alpha);#définition de la fonction
  dq:=simplify(mtaylor(f(x,y),[x=10,y=10],2));#partie
  principale du développement à l'ordre 1
  x,y:=10.1,9.95;#coordonnées du point considéré
  dq;#valeur approchée (à l'ordre 1) de la fonction au point
  considéré
```

$$f := (x, y) \rightarrow x^{\alpha} y^{1-\alpha}$$

$$dq := y - \alpha y + \alpha x$$

$$x, y := 10.1, 9.95$$

$$9.95 + 0.15\,\alpha \qquad (2.1)$$

Mais comme tout est très simple avec Maple, rien n'interdit de demander une approximation plus fine, par exemple quadratique.

```
> restart;
  f:=(x,y)->x^alpha*y^(1-alpha);
  dq:=simplify(mtaylor(f(x,y),[x=10,y=10],3));;#partie
  principale du développement à l'ordre 2
  x,y:=10.1,9.95;
  dq;##valeur approchée (à l'ordre 2) de la fonction au point
  considéré
```

$$f := (x, y) \rightarrow x^{\alpha} y^{1-\alpha}$$

$$dq := y + \alpha x - \alpha y - \frac{1}{20} \alpha y^2 + \frac{1}{20} \alpha^2 y^2 - \frac{1}{20} x^2 \alpha + \frac{1}{20} x^2 \alpha^2 + \frac{1}{10} \alpha x y$$

$$-\frac{1}{10}\alpha^2 x y$$

$$x, y := 10.1, 9.95$$

$$9.95 + 0.14887500\,\alpha + 0.001125000\,\alpha^2 \qquad \textbf{(2.2)}$$

On peut bien entendu augmenter encore l'ordre du développement.

Exercice M3

Enoncé

Calculer la différentielle de $f(x, y, z) = (x^2 + y)\, e^z$ en $M = (a, b, c)$.

Solution

Après avoir rappelé la fonction-procédure Dt exposée dans la section "Développements limités et différentielles", on l'applique à la fonction *f*; puis on attribue aux variables muettes *x*, *y* et *z* les valeurs *a*, *b* et *c* puisque la différentielle doit être calculée au point *M*.

```
> restart;
  Dt:=(f,V)->add(diff(f,v)*cat(d,v),v=V);#appel à la fonction-
  procédure Dt
  df:=Dt((x^2+y)*exp(z),{x,y,z});#utilisation de Dt à la
  fonction demandée
  x,y,z:=a,b,c;#coordonnées du point M
  df;#différentielle de f en M
```

$$Dt := (f, V) \rightarrow add\left(\left(\frac{\partial}{\partial v} f\right) cat(d, v), v = V\right)$$

$$df := 2\, x\, e^z\, dx + e^z\, dy + (x^2 + y)\, e^z\, dz$$

$$x, y, z := a, b, c$$

$$2\, a\, e^c\, dx + e^c\, dy + (a^2 + b)\, e^c\, dz \qquad \textbf{(3.1)}$$

Exercice M4

Enoncé

Soit $f : \mathbb{R}^2 \rightarrow \mathbb{R}$, une fonction de classe C^3 et soit $M = (a, b)$. On considère la fonction d'une variable réelle $g : t \rightarrow g(t) = f(M + t\,H)$ où $H = (u, v)$.
Calculer $g'(t)$ puis $g''(t)$ et $g'''(t)$.
Application : $f(x, y) = x^2\, e^y$, $M = (15; 7)$, $H = (1, 1)$.

Solution

La dérivée première de la fonction *g* est la dérivée de la fonction de deux variables *f* par rapport à *t*.
Il s'agit en fait de pratiquer une dérivation dite "en chaîne", qui ne pose aucun problème à Maple :

```
> restart;
  gp:=diff(f(a+t*u,b+t*v),t);
```

$$gp := D_1(f)(a + t\,u, b + t\,v)\, u + D_2(f)(a + t\,u, b + t\,v)\, v \qquad \textbf{(4.1)}$$

On procède de même pour la dérivée seconde :

```
> gs:=simplify(diff(f(a+t*u,b+t*v),t$2));
```

$$gs := D_{1,1}(f)(a + t\,u, b + t\,v)\, u^2 + 2\, u\, D_{1,2}(f)(a + t\,u, b + t\,v)\, v + D_{2,2}(f)(a + t\,u, b \qquad \textbf{(4.2)}$$

$+ t\, v)\, v^2$

et pour la dérivée troisième :

```
> gt:=simplify(diff(f(a+t*u,b+t*v),t$3));
```

$$gt := \mathrm{D}_{1,\,1,\,1}(f)\,(a + t\,u, b + t\,v)\,u^3 + 3\,u^2\,\mathrm{D}_{1,\,1,\,2}(f)\,(a + t\,u, b + t\,v)\,v + 3\,u\,\mathrm{D}_{1,\,2,\,2}(f)\,(a + t\,u, b + t\,v)\,v^2 + \mathrm{D}_{2,\,2,\,2}(f)\,(a + t\,u, b + t\,v)\,v^3 \quad \textbf{(4.3)}$$

L'application demandée se fait directement en particularisant la fonction et les points.

```
> f:=(x,y)->x^2*exp(y);
  a,b,u,v:=15,7,1,1;
  gp;
  gs;
  gt;
```

$$f := (x, y) \rightarrow x^2\,\mathrm{e}^y$$

$$a, b, u, v := 15, 7, 1, 1$$

$$2\,(15 + t)\,\mathrm{e}^{7+t} + (15 + t)^2\,\mathrm{e}^{7+t}$$

$$2\,\mathrm{e}^{7+t} + 4\,(15 + t)\,\mathrm{e}^{7+t} + (15 + t)^2\,\mathrm{e}^{7+t}$$

$$6\,\mathrm{e}^{7+t} + 6\,(15 + t)\,\mathrm{e}^{7+t} + (15 + t)^2\,\mathrm{e}^{7+t} \quad \textbf{(4.4)}$$

Exercice E1

Enoncé

Dans le Capital de Karl Marx, le taux de profit, *tp*, est égal au rapport de la plus-value absolue, *pl*, et du capital-argent engagé par les capitalistes dans la production, $c + v$, où c exprime la valeur du capital constant et v le capital variable (qui rémunère les travailleurs).
1. On appelle plus-value relative et on note p le rapport de la plus-value absolue et du capital variable. On appelle composition organique du capital et on note k le rapport du capital constant au capital variable. Montrer que le taux de profit est une fonction de p et k.
2. Calculer la différentielle du taux de profit en fonction des différentielles de p et k.

Solution

1. En divisant le numérateur et le dénominateur de $tp = \dfrac{pl}{c + v}$ par v, on aura $tp = \dfrac{\frac{pl}{v}}{\frac{c}{v} + 1} = \dfrac{p}{k + 1}$.

Ainsi, le taux de profit est une fonction rationnelle de deux variables (explicatives) : le taux de plus-value relative et la composition organique du capital.

2. On charge puis utilise la procédure de calcul de différentielle donnée dans le cours.

```
> Dt:=(f,V)->add(diff(f,v)*cat(d,v),v=V):
  dtp:=Dt(p/(k+1),{p,k});
```

$$dtp := -\frac{p\,dk}{(k + 1)^2} + \frac{dp}{k + 1} \quad \textbf{(5.1)}$$

Un accroissement du taux d'exploitation ($dp > 0$) augmente le taux de profit alors qu'un accroissement de la composition organique du capital ($dk > 0$) le réduit.

Exercice E2

Enoncé

Une fonction de production dite CES (Constant Elasticity of Substitution) s'écrit :

$$q(x, y) = A\left[\alpha\, x^{-\rho} + (1-\alpha)\, y^{-\rho}\right]^{-\frac{1}{\rho}}$$

où x et y représentent deux facteurs de production, A est un facteur d'échelle (réel strictement positif), α est un réel strictement compris entre 0 et 1 et ρ est un paramètre réel.

1. Montrer qu'une fonction Cobb-Douglas est une fonction CES pour laquelle $\rho = 0$.

2. Calculer le taux de substitution technique $\dfrac{\frac{\partial q}{\partial x}(x, y)}{\frac{\partial q}{\partial y}(x, y)}$ puis montrer qu'il est décroissant par rapport à x si $\rho > -1$.

3. Calculer la matrice hessienne d'une CES par rapport aux deux facteurs de production puis son déterminant.

Solution

Commençons par poser l'expression d'une fonction CES ainsi que les hypothèses sur les paramètres.

```
> restart;
  Xpces:=A*(alpha*x^(-rho)+(1-alpha)*y^(-rho))^(-1/rho);
  assume(A>0,alpha>0,alpha<1);
```

$$Xpces := A\left(\alpha\, x^{-\rho} + (1-\alpha)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \qquad \textbf{(6.1)}$$

1. Considérons la CES comme une fonction de ρ :

```
> cesrho:=unapply(Xpces,rho);
```

$$cesrho := \rho \to A\sim \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \qquad \textbf{(6.2)}$$

Elle n'est pas définie en $\rho = 0$ (à cause de l'exposant $-\dfrac{1}{\rho}$).

```
> cesrho(0);
Error, (in cesrho) numeric exception: division by zero
```

Mais rien n'interdit de prolonger la fonction par continuité. Calculons alors la limite de la CES quand ρ tend vers 0.

```
> limit(cesrho(rho),rho=0);
  simplify(%,power);
```

$$\frac{x^{\alpha\sim}\, y\, A\sim}{y^{\alpha\sim}}$$

$$x^{\alpha\sim}\, y^{1-\alpha\sim}\, A\sim \qquad \textbf{(6.3)}$$

A la limite, c'est à dire quand le paramètre ρ tend vers 0, une CES devient une fonction de production Cobb-Douglas.

2. Considérons à présent la CES comme une fonction des deux variables x et y.

```
> ces:=unapply(Xpces,(x,y));
```

(6.4)

$$ces := (x, y) \rightarrow A\sim \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \tag{6.4}$$

Le taux de substitution technique est le rapport des dérivées partielles premières. Il est lui-même une fonction de *x* et de *y*.

```
> tst:=simplify(D[1](ces)(x,y)/D[2](ces)(x,y));
```

$$tst := -\frac{\alpha\sim x^{-\rho-1}\, y^{\rho+1}}{-1+\alpha\sim} \tag{6.5}$$

On doit montrer que les TST décroît quand *x* augmente. Calculons sa dérivée par rapport à *x* pour étudier son signe.

```
> tstx:=simplify(diff(tst,x));
```

$$tstx := \frac{\alpha\sim x^{-\rho-2}\,(\rho+1)\, y^{\rho+1}}{-1+\alpha\sim} \tag{6.6}$$

Comme *x* et *y* représentent des facteurs de production, ils prennent des valeurs strictement positives, de sorte que $x^{-\rho-2} y^{\rho+1}$ est positif. Le test suivant confirme alors que la dérivée du TST par rapport à *x* est strictement négative dans l'hypothèse où ρ est strictement supérieur à 1.

```
> is(alpha*(1+rho)/(-1+alpha)<0) assuming rho>-1;
```

$$true \tag{6.7}$$

3. Pour calculer le déterminant de la matrice hessienne avec la commande **Determinant** du paquetage **LinearAlgebra**, celle-ci doit être du type *Matrix*. Le plus simple est de charger le paquetage **VectorCalculus** pour accéder à la commande **Hessian**.

```
> with(VectorCalculus):
  D2ces:=Hessian(Xpces,[x,y]);
  whattype(D2ces);
```

$$D2ces := \Bigg[\Bigg[\frac{A\sim \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \alpha\sim^2 \left(x^{-\rho}\right)^2}{x^2 \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)^2}$$

$$- \frac{A\sim \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \alpha\sim x^{-\rho} \rho}{x^2 \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)}$$

$$- \frac{A\sim \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \alpha\sim x^{-\rho}}{x^2 \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)}$$

$$+ \frac{A\sim \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \alpha\sim^2 \left(x^{-\rho}\right)^2 \rho}{x^2 \left(\alpha\sim x^{-\rho} + (1-\alpha\sim)\, y^{-\rho}\right)^2},$$

$$\frac{A\sim \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} (1-\alpha\sim)\, y^{-\rho}\, \alpha\sim x^{-\rho}}{y \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{2} x}$$

$$\left. + \frac{A\sim \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \alpha\sim x^{-\rho} (1-\alpha\sim)\, y^{-\rho} \rho}{x \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{2} y} \right],$$

$$\left[\frac{A\sim \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} (1-\alpha\sim)\, y^{-\rho}\, \alpha\sim x^{-\rho}}{y \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{2} x} \right.$$

$$+ \frac{A\sim \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} \alpha\sim x^{-\rho} (1-\alpha\sim)\, y^{-\rho} \rho}{x \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{2} y},$$

$$\frac{A\sim \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} (1-\alpha\sim)^{2} \left(y^{-\rho}\right)^{2}}{y^{2} \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{2}}$$

$$- \frac{A\sim \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} (1-\alpha\sim)\, y^{-\rho} \rho}{y^{2} \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)}$$

$$- \frac{A\sim \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} (1-\alpha\sim)\, y^{-\rho}}{y^{2} \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)}$$

$$\left. + \frac{A\sim \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{-\frac{1}{\rho}} (1-\alpha\sim)^{2} \left(y^{-\rho}\right)^{2} \rho}{y^{2} \left(\alpha\sim x^{-\rho}+(1-\alpha\sim)\, y^{-\rho}\right)^{2}} \right]\Bigg]$$

$$Matrix \qquad (6.8)$$

```
> with(LinearAlgebra):
  Determinant(D2ces);
```

$$0 \qquad (6.9)$$